
1. Overview

Verification Test Summary

Test Case:
TC_24_1_2
Test Group:
LTE_A_PRO_Testsuite
ATS Version:
iwd-TTCN3-B2018-03_D18wk24 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Qualcomm MDM 9150
Verification Status:
PASS
3. Corrections required

3.1. Correction to function TC_24_1_2() 
	Function name
	TC_24_1_2()

	Reason for change
	The guard time is not sufficient to ensure the test is run to completion

	Summary of change
	Increased the guard time to 720 sec

	TTCN module
	V2V_Communication_Testcases_SideLink.ttcn

	MCC160 Comment
	Accepted.


Before Change:

	  testcase TC_24_1_2() runs on MTC_LTE_V2X system SYSTEM_LTE_V2X {

    // @purpose

    //    V2X Sidelink Communication / Pre-configured authorisation / Utilisation of the pre-configured resources / Transmission

    var EUTRA_V2X_PTC    v_EUTRA      := null;

    var SideLink_PTC     v_SideLink   := null;

    timer t_GuardTimer := int2float(360);

  


After Change:

	  testcase TC_24_1_2() runs on MTC_LTE_V2X system SYSTEM_LTE_V2X {

    // @purpose

    //    V2X Sidelink Communication / Pre-configured authorisation / Utilisation of the pre-configured resources / Transmission

    var EUTRA_V2X_PTC    v_EUTRA      := null;

    var SideLink_PTC     v_SideLink   := null;

    timer t_GuardTimer := int2float(720);

  


3.2. Correction to function f_TC_24_1_2_SideLink
	Function name
	f_TC_24_1_2_SideLink

	Reason for change
	1. Extra Co-ord message from sidelink PTC triggers EUTRA_SL PTC to signal GNSS simulator to start move scenario before the STCH PDCP SDU is received from the UE. Reception of STCH PDCP SDU by the UE is the indication that UE has aquired GNSS signal and the move should not be trigerred before that. 

	Summary of change
	1.  Removed the extra co-ord message.

	TTCN module
	V2V_Communication_Testcases_SideLink.ttcn

	MCC160 Comment
	Accepted.


Before Change:

	function f_TC_24_1_2_SideLink() runs on SideLink_PTC

  {

    var SideLink_Coordination_MSG v_SideLink_OutOfCoverage;

    var template (value) SL_CommResourcePoolV2X_r14 v_SL_CommResourcePoolV2X_Rx_PSSCH;

    var SL_SizeSubchannel_r14_Type v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertDL_BandwidthToInteger(px_eSecondaryBandChannelBandwidth));

    var default v_DefaultRef;

    // Wait for coordination with EUTRA PTC

    v_SideLink_OutOfCoverage := f_SideLink_WaitForCoordMsg(EUTRA, cmr_SideLink_OutOfCoverageInfo);

    //Configure SS-UE1 in Rx for UE Tx using Preconfigured information, UTC Synchronised

    f_SSUE_V2X_Config(ss_UE_Id1,

                      UTC_Synchronised,

                      v_SideLink_OutOfCoverage,

                      SSUE_Rx,

                      tsc_V2X_BITMAP_6 // According to 36.508 Table 4.6.3-20K

                      );

    //SSUE-1 ready for start of TC

    f_SideLink_SendCoordMsg(EUTRA);
    f_SideLink_TestBody_Set(true);

    //UE may start transmitting

    f_SideLink_WaitForCoordMsg(EUTRA);

    v_DefaultRef := activate(a_SidelinkUE_STCH_Reception(ss_UE_Id1));

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit V2X sidelink communication data over the PC5 interface  in accordance with the

    //       resources pre-configured in SL-V2X-Preconfiguration-r14 -> v2x-PreconfigFreqList-r14"?

    v_SL_CommResourcePoolV2X_Rx_PSSCH := cs_SL_CommResourcePoolV2X (tsc_V2X_BITMAP_6, v_SizeSubchannel);

    f_CheckCommDataReception_V2X(ss_UE_Id1, "Step 5", v_SL_CommResourcePoolV2X_Rx_PSSCH, UTC_Synchronised);

    f_SideLink_SendCoordMsg(EUTRA);

    f_SideLink_WaitForCoordMsg(EUTRA);

    deactivate(v_DefaultRef);

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit, in the next 10 seconds, V2X sidelink communication data over the PC5 interface in accordance with the

    //       resources pre-configured in SL-V2X-Preconfiguration-r14 -> v2x-PreconfigFreqList-r14"?

    f_SSUE_CheckNoSTCH_DataReceived(ss_UE_Id1, "Step 8", 10.0);

    f_SideLink_SendCoordMsg(EUTRA);

    f_SideLink_TestBody_Set(false);

    //Release SS-UE1

    f_SSUE_ReleaseAll();

  }


After Change:

	function f_TC_24_1_2_SideLink() runs on SideLink_PTC

  {

    var SideLink_Coordination_MSG v_SideLink_OutOfCoverage;

    var template (value) SL_CommResourcePoolV2X_r14 v_SL_CommResourcePoolV2X_Rx_PSSCH;

    var SL_SizeSubchannel_r14_Type v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertDL_BandwidthToInteger(px_eSecondaryBandChannelBandwidth));

    var default v_DefaultRef;

    // Wait for coordination with EUTRA PTC

    v_SideLink_OutOfCoverage := f_SideLink_WaitForCoordMsg(EUTRA, cmr_SideLink_OutOfCoverageInfo);

    //Configure SS-UE1 in Rx for UE Tx using Preconfigured information, UTC Synchronised

    f_SSUE_V2X_Config(ss_UE_Id1,

                      UTC_Synchronised,

                      v_SideLink_OutOfCoverage,

                      SSUE_Rx,

                      tsc_V2X_BITMAP_6 // According to 36.508 Table 4.6.3-20K

                      );

    //SSUE-1 ready for start of TC

    //f_SideLink_SendCoordMsg(EUTRA);
    f_SideLink_TestBody_Set(true);

    //UE may start transmitting

    f_SideLink_WaitForCoordMsg(EUTRA);

    v_DefaultRef := activate(a_SidelinkUE_STCH_Reception(ss_UE_Id1));

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit V2X sidelink communication data over the PC5 interface  in accordance with the

    //       resources pre-configured in SL-V2X-Preconfiguration-r14 -> v2x-PreconfigFreqList-r14"?

    v_SL_CommResourcePoolV2X_Rx_PSSCH := cs_SL_CommResourcePoolV2X (tsc_V2X_BITMAP_6, v_SizeSubchannel);

    f_CheckCommDataReception_V2X(ss_UE_Id1, "Step 5", v_SL_CommResourcePoolV2X_Rx_PSSCH, UTC_Synchronised);

    f_SideLink_SendCoordMsg(EUTRA);

    f_SideLink_WaitForCoordMsg(EUTRA);

    deactivate(v_DefaultRef);

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit, in the next 10 seconds, V2X sidelink communication data over the PC5 interface in accordance with the

    //       resources pre-configured in SL-V2X-Preconfiguration-r14 -> v2x-PreconfigFreqList-r14"?

    f_SSUE_CheckNoSTCH_DataReceived(ss_UE_Id1, "Step 8", 10.0);

    f_SideLink_SendCoordMsg(EUTRA);

    f_SideLink_TestBody_Set(false);

    //Release SS-UE1

    f_SSUE_ReleaseAll();

  }


3.3. Additional common changes

Additional common changes have been discussed with MCC160 and implemented based on MCC160 comments. These changes are not documented in this document. The changes are mainly in.

f_MTC_ConnectPTCs_LTE_V2X

f_MTC_ConnectPTCs_LTE_SL_Common

f_GetTestcaseAttrib_Specific_OutOfCoverage_TC

cs_SSUE_ReqAspCommonPart /cs_SSUE_ReqAspCommonPart_Cfg / cas_SSUE_Config

f_SL_CheckIsSTCH_ReceivedInCorrectSubframe

MCC160> For clarification, the additional changes are added below in clause 3.3.1 to 3.3.7
3.3.1
Correction to function f_MTC_ConnectPTCs_LTE_V2X

	Function name
	f_MTC_ConnectPTCs_LTE_V2X

	Reason for change
	TTCN creates and maps EUTRA and NASEMU PTC for Out of coverage TCs which is incorrect and not necessary. This can create warning messages if SS is not initializing LTE components of Out of coverage TCs. Also Connection of MTC's system ports is missing

	Summary of change
	Made correction to ttcn to not map EUTRA and NASEMU PTC for Out of coverage TCs

	TTCN module
	MTC_Main_LTE_V2X.ttcn

	MCC160 Comment
	Change 1 – add f_MTC_BASE_Map(p_System) to map MTC’s system port – accepted

Change 2 – implemented differently: inside function f_MTC_ConnectPTCs_LTE_SL_Common – see proposed implementation below and in 3.3.2


Before Change:

	function f_MTC_ConnectPTCs_LTE_V2X(SYSTEM_LTE_V2X p_System,
                                     EUTRA_V2X_PTC p_Eutra,
                                     SideLink_PTC p_SideLink) runs on MTC_LTE_V2X
  {
    var boolean v_MapPOSSYS := f_GetTestcaseAttrib_SpecificMTC_SatPort(testcasename());
    f_MTC_ConnectPTCs_LTE_SL_Common(p_System, p_Eutra, p_SideLink);
    if (v_MapPOSSYS) {
      map(mtc:POSSYS, p_System:POSSYS);
      connect(mtc:EUTRA_POSSYS, p_Eutra:POSSYS);
    }
    f_UT_InitialPowerOffUE(Ut);
  }


MCC160 proposed implementation:
	  function f_MTC_ConnectPTCs_LTE_V2X(SYSTEM_LTE_V2X p_System,

                                     EUTRA_V2X_PTC p_Eutra,

                                     SideLink_PTC p_SideLink) runs on MTC_LTE_V2X

  {

    var boolean v_MapPOSSYS := f_GetTestcaseAttrib_SpecificMTC_SatPort(testcasename());

    // ***** MTC *****

    f_MTC_BASE_Map(p_System);    // Map MTC's system ports
    f_MTC_ConnectPTCs_LTE_SL_Common(p_System, p_Eutra, p_SideLink);

    if (v_MapPOSSYS) {

      map(mtc:POSSYS, p_System:POSSYS);

      connect(mtc:EUTRA_POSSYS, p_Eutra:POSSYS);

    }

    f_UT_InitialPowerOffUE(Ut);

  }


3.3.2
Correction to function f_MTC_ConnectPTCs_LTE_SL_Common

	Function name
	f_MTC_ConnectPTCs_LTE_SL_Common

	Reason for change
	TTCN creates and maps EUTRA and NASEMU PTC for Out of coverage TCs which is incorrect and not necessary. This can create warning messages if SS is not initializing LTE components of Out of coverage TCs

	Summary of change
	Made correction to ttcn to not map EUTRA and NASEMU PTC for Out of coverage TCs

	TTCN module
	MTC_Main_LTE_SL.ttcn

	MCC160 Comment
	Accepted, see proposed implementation below


Before Change:

	function f_MTC_ConnectPTCs_LTE_SL_Common(SYSTEM_LTE_SL p_System,
                                           EUTRA_SL_PTC p_Eutra,
                                           SideLink_PTC p_SideLink,
                                           boolean p_DnsRequired := false,
                                           boolean p_HttpRequired := false,
                                           boolean p_UdpClientRequired := false) runs on MTC_LTE_SL
  {
    var IMS_PTC v_IMS_PTC;
    var NASEMU_EUTRA_PTC v_NASEMU_PTC;
    vc_Components.Eutra := p_Eutra;
    if (p_Eutra == null) {
      FatalError (__FILE__, __LINE__, "invalid configuration");
      mtc.stop;
    }
    // ***** EUTRA PTC *****
    f_EUTRA_PTC_Map(p_System, p_Eutra);                                     // map EUTRA system ports
    v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap_EUTRA(p_System, p_Eutra);  // create NasEmu, map system ports and connect with EUTRA PTC
    v_NASEMU_PTC.start(f_NASEMU_MainLoop_EUTRA());                          // start NASEMU
    vc_IP_PTC := f_IP_PTC_CreateAndMap(p_System);
    connect(p_Eutra:IP, vc_IP_PTC:EUTRA_CTRL);
    vc_IP_PTC.start(f_IP_PTC_MainLoop(p_DnsRequired, p_HttpRequired, p_UdpClientRequired));
    // ***** Create, connect and start IMS PTC *****
    v_IMS_PTC := f_IMS_PTC_Create();
    connect(v_IMS_PTC:IMS_CTRL,      vc_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);
    connect(v_IMS_PTC:IMS_Server,    vc_IP_PTC:IMS_Server[tsc_Index_PDN1]);
    connect(v_IMS_PTC:IMS_Client,    vc_IP_PTC:IMS_Client[tsc_Index_PDN1]);
    connect(v_IMS_PTC:IPCAN,         p_Eutra:IMS[tsc_Index_PDN1]);
    v_IMS_PTC.start(f_IMS_DefaultHandler());
    // ***** Connect EUTRA to MTC *****
    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA], p_Eutra:UT);
    // ***** Connect SIDELINK *****
    if (p_SideLink != null) {
      connect(p_SideLink:EUTRA,  p_Eutra:SIDELINK);
      f_SideLink_PTC_Map(p_System, p_SideLink);
    }
  }


MCC160 proposed implemenation
  function f_MTC_ConnectPTCs_LTE_SL_Common(SYSTEM_LTE_SL p_System,

                                           EUTRA_SL_PTC p_Eutra,

                                           SideLink_PTC p_SideLink,

                                           boolean p_DnsRequired := false,

                                           boolean p_HttpRequired := false,

                                           boolean p_UdpClientRequired := false) runs on MTC_LTE_SL

  {

    var IMS_PTC v_IMS_PTC;

    var NASEMU_EUTRA_PTC v_NASEMU_PTC;

    var boolean v_NotMapLTESYS := f_GetTestcaseAttrib_SpecificOutOfCoverage(testcasename()); 

    vc_Components.Eutra := p_Eutra;

    if (p_Eutra == null) {

      FatalError (__FILE__, __LINE__, "invalid configuration");

      mtc.stop;

    }

    // ***** EUTRA PTC *****

    if (not v_NotMapLTESYS) {    

        f_EUTRA_PTC_Map(p_System, p_Eutra);                                     // map EUTRA system ports

        v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap_EUTRA(p_System, p_Eutra);  // create NasEmu, map system ports and connect with EUTRA PTC

        v_NASEMU_PTC.start(f_NASEMU_MainLoop_EUTRA());                          // start NASEMU

        vc_IP_PTC := f_IP_PTC_CreateAndMap(p_System);

        connect(p_Eutra:IP, vc_IP_PTC:EUTRA_CTRL);

        vc_IP_PTC.start(f_IP_PTC_MainLoop(p_DnsRequired, p_HttpRequired, p_UdpClientRequired));

        // ***** Create, connect and start IMS PTC *****

        v_IMS_PTC := f_IMS_PTC_Create();

        connect(v_IMS_PTC:IMS_CTRL,      vc_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);

        connect(v_IMS_PTC:IMS_Server,    vc_IP_PTC:IMS_Server[tsc_Index_PDN1]);

        connect(v_IMS_PTC:IMS_Client,    vc_IP_PTC:IMS_Client[tsc_Index_PDN1]);

        connect(v_IMS_PTC:IPCAN,         p_Eutra:IMS[tsc_Index_PDN1]);

        v_IMS_PTC.start(f_IMS_DefaultHandler());

    }
    // ***** Connect EUTRA to MTC *****

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA], p_Eutra:UT);

    // ***** Connect SIDELINK *****

    if (p_SideLink != null) {

      connect(p_SideLink:EUTRA,  p_Eutra:SIDELINK);

      f_SideLink_PTC_Map(p_System, p_SideLink);

    }

  }

3.3.3
Addition of function f_GetTestcaseAttrib_Specific_OutOfCoverage_TC

	Function name
	f_GetTestcaseAttrib_Specific_OutOfCoverage_TC

	Reason for change
	Function listing Out of coverage TCs is missing in ttcn.

Note : Linked to changes 3.3.1 and 3.3.2

	Summary of change
	Added function to list Out of coverage TCs

	TTCN module
	\LTE_A_PRO\common\TestcaseProperties.ttcn

	MCC160 Comment
	Accepted, rename the attribute to SpecificOutOfCoverage


After Change:

	function f_GetTestcaseAttrib_Specific_OutOfCoverage_TC(charstring p_Testcase) return boolean
  {
    select (p_Testcase) {
      case ("TC_24_1_16") { return true; }
      case ("TC_24_1_19") { return true; }
      case ("TC_24_1_2") { return true; }
      case ("TC_24_1_4") { return true; }
    }
    return false;
  }


3.3.4
cs_SSUE_ReqAspCommonPart /cs_SSUE_ReqAspCommonPart_Cfg / cas_SSUE_Config

	Template name
	cs_SSUE_ReqAspCommonPart / cs_SSUE_ReqAspCommonPart_Cfg / cas_SSUE_Config

	Reason for change
	TTCN configures Tx data (e.g. congestion) in parallel with the SUE configuration that can lead to race condition where the SUE is not yet ready for transmission.

	Summary of change
	switching on the CNF flag for the SUE configuration that ensures through the handshake that the SUE is ready before the Tx data / congestion is started.

	TTCN module
	\LTE_A_PRO\common\ SideLink_CommonFunctions.ttcn

	MCC160 Comment
	Accepted in principle but implemented with the default value set to false in cas_SSUE_Config, the parameter is then set to true for the reconfiguration before transmitting congestion. 


Before Change:

	  template (value) SL_ReqAspCommonPart_Type cs_SSUE_ReqAspCommonPart(SS_UE_Id_Type p_SSUE_Id,

                                                                     template (value) SL_RoutingInfo_Type p_SL_RoutingInfo,

                                                                     template (value) SL_TimingInfo_Type p_SL_TimingInfo) :=

  {

    SS_UE_Id := p_SSUE_Id,

    RoutingInfo := p_SL_RoutingInfo,

    TimingInfo := p_SL_TimingInfo,

    ControlInfo := {

      CnfFlag      := false,

      FollowOnFlag := false

    }

  };

…
template (value) SL_ReqAspCommonPart_Type cs_SSUE_ReqAspCommonPart_Cfg(

                              SS_UE_Id_Type p_SSUE_Id,

                              template (value) SL_TimingInfo_Type p_SL_TimingInfo) :=     

cs_SSUE_ReqAspCommonPart(p_SSUE_Id, cs_SL_RoutingInfo_None, p_SL_TimingInfo);

….
template (value) SL_SYSTEM_CTRL_REQ cas_SSUE_Config(

                   SS_UE_Id_Type p_SSUE_Id,

                   template (value) SL_TimingInfo_Type p_SL_TimingInfo,

                   template (value) SS_UE_ConfigInfo_Type p_SS_UE_ConfigInfo_Type) :=

  {

    Common := cs_SSUE_ReqAspCommonPart_Cfg(p_SSUE_Id, p_SL_TimingInfo),

    Request := {

      SS_UE_Config := {

        AddOrReconfigure := p_SS_UE_ConfigInfo_Type

      }

    }

  };


MCC160 proposed implementation:
template (value) SL_SYSTEM_CTRL_REQ cas_SSUE_Config(SS_UE_Id_Type p_SSUE_Id,

                                     template (value) SL_TimingInfo_Type p_SL_TimingInfo,

                                     template (value) SS_UE_ConfigInfo_Type p_SS_UE_ConfigInfo_Type,
                                     boolean p_CnfFlag := false
                                                      ) :=

  {

    Common := cs_SSUE_ReqAspCommonPart_Cfg(p_SSUE_Id, p_SL_TimingInfo, p_CnfFlag),

    Request := {

      SS_UE_Config := {

        AddOrReconfigure := p_SS_UE_ConfigInfo_Type

      }

    }

  };

function f_SSUE_CommonReConfigV2X ( SS_UE_Id_Type p_SSUE_Id,

                   template (value) SL_TimingInfo_Type p_SL_TimingInfo,

                    template (omit) SS_UE_CoverageSynchronisation_Type p_SSUE_ProSeCoverage := omit,

                    template (omit) V2X_CommConfig_Type p_V2X_CommConfig := omit,

                    template (omit) SS_UE_Power_Type p_SS_UE_Power := omit,
                    boolean p_CnfFlag := false
                    ) runs on SideLink_PTC

  {

    f_SSUE_CommonConfig(p_SSUE_Id,

                        cas_SSUE_Config(p_SSUE_Id,

                                        p_SL_TimingInfo,

                                        cs_SS_UE_V2X_Reconfig(p_SSUE_ProSeCoverage,

                                                              p_V2X_CommConfig,

                                                              p_SS_UE_Power

                                                              ),
                                        p_CnfFlag
                        ));

   }

3.3.5
Correction to f_V2X_SendOutOfConverageInfo
	Template name
	f_V2X_SendOutOfConverageInfo

	Reason for change
	FDD mode is hardcoded in the coordination message sent from E-UTRA PTC to Sidelink PTC. The mode shall be retreived from the secondary band under test and pass to the coordination message.

	Summary of change
	Add variable v_EUTRA_FDD_TDD_Mode initialized with TDD

Assign v_EUTRA_FDD_TDD_Mode to FDD if px_eSecondaryFrequencyBand is FDD

	TTCN module
	EUTRA_V2X_CommonFunctions

	MCC160 Comment
	Accepted.


Before Change:

	function f_V2X_SendOutOfConverageInfo(Dl_Bandwidth_Type p_DL_Bandwidth := px_eSecondaryBandChannelBandwidth) runs on EUTRA_SL_PTC

  {

    var Ul_Bandwidth_Type  v_UL_ChBandwidth := f_ConvertDL_BandwidthToUL(p_DL_Bandwidth);

    var Frequency_fList_Type v_Freq := f_EUTRA_InitFrequencyV2X_Communication(px_eSecondaryFrequencyBand, v_UL_ChBandwidth);

    f_SideLink_SendCoordMsg(SIDELINK, cms_SideLink_OutOfCoverageInfo(v_Freq.f1.ul_CarrierFreq_r9, v_UL_ChBandwidth, FDD));

  } 


After Change:

	  function f_V2X_SendOutOfConverageInfo(Dl_Bandwidth_Type p_DL_Bandwidth := px_eSecondaryBandChannelBandwidth) runs on EUTRA_SL_PTC

  {

    var Ul_Bandwidth_Type  v_UL_ChBandwidth := f_ConvertDL_BandwidthToUL(p_DL_Bandwidth);

    var Frequency_fList_Type v_Freq := f_EUTRA_InitFrequencyV2X_Communication(px_eSecondaryFrequencyBand, v_UL_ChBandwidth);

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode := TDD;
    if (f_EUTRA_FrequencyBandIsFDD(px_eSecondaryFrequencyBand)){

        v_EUTRA_FDD_TDD_Mode := FDD;

    } 

    f_SideLink_SendCoordMsg(SIDELINK, cms_SideLink_OutOfCoverageInfo(v_Freq.f1.ul_CarrierFreq_r9, v_UL_ChBandwidth, v_EUTRA_FDD_TDD_Mode));

  }


3.3.6
Correction to f_SSUE_V2X_Config
	

	Template name
	f_SSUE_V2X_Config

	Reason for change
	The variable v_SL_TimingInfo is set to cs_TimingIC_Now for all coverage type including UTC synchronised. But SS-UE shall be configured with the correct timing.

	Summary of change
	Initialise v_SL_TimingInfo according to the coverage type.

	TTCN module
	SideLink_V2X_CommonFunctions

	MCC160 Comment
	Accepted.


Before Change:

	  function f_SSUE_V2X_Config(SS_UE_Id_Type              p_SSUE_Id,

                             SSUE_Coverage_Type         p_Coverage_Type,

                             SideLink_Coordination_MSG  p_SL_Info,

                             SL_DC_UE_Type              p_V2X_DC_UE_Type,

                             template(omit)  B20_Type   p_RxBitmap := omit,

                             template(omit)  B20_Type   p_TxBitmap := omit,

                             AbsoluteCellPower_Type     p_AbsoluteCellPower := tsc_SS_UE_Serving,

                             template(omit) V2X_CommSyncConfig_Type p_V2X_CommSyncConfig := omit,

                             B24_Type p_SSUE_Layer2Id := tsc_SS_UE1_Layer2Id

                             )

    runs on SideLink_PTC

  {

    var template (value) SL_TimingInfo_Type v_SL_TimingInfo;

    var template (value) SS_UE_CoverageSynchronisation_Type v_SSUE_ProSeCoverage;

    var template (value) V2X_CommConfig_Type v_V2X_CommConfig;

    var SL_SizeSubchannel_r14_Type v_SizeSubchannel;

    var Ul_Bandwidth_Type v_SidelinkBandwidth;

    v_SL_TimingInfo := cs_TimingIC_Now;

    select (p_Coverage_Type) {

      case(InCoverage){

        v_SSUE_ProSeCoverage := cs_SSUE_IC(p_SL_Info.InCoverageInfo.CellInCoverageInfo);

        v_SidelinkBandwidth := p_SL_Info.InCoverageInfo.CellInCoverageInfo.UL_Bandwidth;

        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(v_SidelinkBandwidth));

      }

      case(UTC_Synchronised){

        v_SSUE_ProSeCoverage := cs_SSUE_UTC(p_SL_Info.OutOfCoverageInfo);

        v_SidelinkBandwidth := p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth;

        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(v_SidelinkBandwidth));

      }

      case(OutOfCoverage) {

        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth));

        v_SidelinkBandwidth := p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth;

        v_SSUE_ProSeCoverage := cs_SSUE_OoC_Common(p_SL_Info.OutOfCoverageInfo.CarrierFreq_r12,

                                                   f_ConvertBandwidth_ToSL_Type(v_SidelinkBandwidth),

                                                   cs_TimingInfo_SSUE1_Def);

      }

      case else {FatalError(__FILE__, __LINE__, "Invalid Coverage type");}

    }

  ….

}  


After Change:

	  function f_SSUE_V2X_Config(SS_UE_Id_Type              p_SSUE_Id,

                             SSUE_Coverage_Type         p_Coverage_Type,

                             SideLink_Coordination_MSG  p_SL_Info,

                             SL_DC_UE_Type              p_V2X_DC_UE_Type,

                             template(omit)  B20_Type   p_RxBitmap := omit,

                             template(omit)  B20_Type   p_TxBitmap := omit,

                             AbsoluteCellPower_Type     p_AbsoluteCellPower := tsc_SS_UE_Serving,

                             template(omit) V2X_CommSyncConfig_Type p_V2X_CommSyncConfig := omit,

                             B24_Type p_SSUE_Layer2Id := tsc_SS_UE1_Layer2Id

                             )

    runs on SideLink_PTC

  {

    var template (value) SL_TimingInfo_Type v_SL_TimingInfo;

    var template (value) SS_UE_CoverageSynchronisation_Type v_SSUE_ProSeCoverage;

    var template (value) V2X_CommConfig_Type v_V2X_CommConfig;

    var SL_SizeSubchannel_r14_Type v_SizeSubchannel;

    var Ul_Bandwidth_Type v_SidelinkBandwidth;

    select (p_Coverage_Type) {

      case(InCoverage){

        v_SL_TimingInfo := cs_TimingIC_Now;
        v_SSUE_ProSeCoverage := cs_SSUE_IC(p_SL_Info.InCoverageInfo.CellInCoverageInfo);

        v_SidelinkBandwidth := p_SL_Info.InCoverageInfo.CellInCoverageInfo.UL_Bandwidth;

        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(v_SidelinkBandwidth));

      }

      case(UTC_Synchronised){

        v_SL_TimingInfo := cs_TimingUTC_Now;
        v_SSUE_ProSeCoverage := cs_SSUE_UTC(p_SL_Info.OutOfCoverageInfo);

        v_SidelinkBandwidth := p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth;

        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(v_SidelinkBandwidth));

      }

      case(OutOfCoverage) {

        v_SL_TimingInfo := cs_TimingOC_Now;
        v_SizeSubchannel := f_V2X_ReturnSL_SizeSubchannel(f_ConvertUL_BandwidthToInteger(p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth));

        v_SidelinkBandwidth := p_SL_Info.OutOfCoverageInfo.Ul_Bandwidth;

        v_SSUE_ProSeCoverage := cs_SSUE_OoC_Common(p_SL_Info.OutOfCoverageInfo.CarrierFreq_r12,

                                                   f_ConvertBandwidth_ToSL_Type(v_SidelinkBandwidth),

                                                   cs_TimingInfo_SSUE1_Def);

      }

…

}


3.3.7. 
Correction to function f_SL_CheckIsSTCH_ReceivedInCorrectSubframe
	Template name
	f_SL_CheckIsSTCH_ReceivedInCorrectSubframe

	Reason for change
	Calcutation for v_RxAbsoluteSubframe for case UTC_Synchronised and OutOfCoverage results in a runtime error

	Summary of change
	Corrected ttcn for case UTC_Synchronised and OutOfCoverage

	TTCN module
	SideLink_V2X_CommonFunctions.ttcn

	MCC160 Comment
	Accepted


Before Change:

	function f_SL_CheckIsSTCH_ReceivedInCorrectSubframe( SL_CommResourcePoolV2X_r14 p_SL_CommResourcePoolV2X_Rx,
                                                       SL_TimingInfo_Type p_Timing,
                                                       SSUE_Coverage_Type p_Coverage_Type
                                                      ) return boolean
  {
    var boolean v_Result;
    var integer v_RxAbsoluteSubframe;
    var integer v_RxPositionInBitmap;
    var integer v_OffsetIndicator:= 0; // According to SL-CommResourcePoolV2X-r14-DEFAULT
    var integer v_BitmapLength := 20; // According to SL-CommResourcePoolV2X-r14-DEFAULT
    //Check in time domain
    select (p_Coverage_Type) {
        case(InCoverage){
            v_RxAbsoluteSubframe := p_Timing.InCoverageTiming.SubFrame.SFN.Number*10 + p_Timing.InCoverageTiming.SubFrame.Subframe.Number;
        }
        case(UTC_Synchronised){
            v_RxAbsoluteSubframe := bit2int(p_Timing.UTC_SynchronisedTiming.DirectSubFrame.DFN)*10 + p_Timing.OutOfCoverageTiming.DirectSubFrame.DSubframe;
        }
        case(OutOfCoverage) {
            v_RxAbsoluteSubframe := bit2int(p_Timing.OutOfCoverageTiming.DirectSubFrame.DFN)*10 + p_Timing.UTC_SynchronisedTiming.DirectSubFrame.DSubframe;
        }
        case else {FatalError(__FILE__, __LINE__, "Invalid Coverage type");}
    }
    v_RxPositionInBitmap := (v_RxAbsoluteSubframe - v_OffsetIndicator) mod v_BitmapLength;
    v_Result := (p_SL_CommResourcePoolV2X_Rx.sl_Subframe_r14.bs20_r14[v_RxPositionInBitmap] == '1'B);
    return v_Result;
  }


After Change:

	function f_SL_CheckIsSTCH_ReceivedInCorrectSubframe( SL_CommResourcePoolV2X_r14 p_SL_CommResourcePoolV2X_Rx,
                                                       SL_TimingInfo_Type p_Timing,
                                                       SSUE_Coverage_Type p_Coverage_Type
                                                      ) return boolean
  {
    var boolean v_Result;
    var integer v_RxAbsoluteSubframe;
    var integer v_RxPositionInBitmap;
    var integer v_OffsetIndicator:= 0; // According to SL-CommResourcePoolV2X-r14-DEFAULT
    var integer v_BitmapLength := 20; // According to SL-CommResourcePoolV2X-r14-DEFAULT
    //Check in time domain
    select (p_Coverage_Type) {
        case(InCoverage){
            v_RxAbsoluteSubframe := p_Timing.InCoverageTiming.SubFrame.SFN.Number*10 + p_Timing.InCoverageTiming.SubFrame.Subframe.Number;
        }
        case(UTC_Synchronised){
            v_RxAbsoluteSubframe := bit2int(p_Timing.UTC_SynchronisedTiming.DirectSubFrame.DFN)*10 + p_Timing.UTC_SynchronisedTiming.DirectSubFrame.DSubframe;
        }
        case(OutOfCoverage) {
            v_RxAbsoluteSubframe := bit2int(p_Timing.OutOfCoverageTiming.DirectSubFrame.DFN)*10 + p_Timing.OutOfCoverageTiming.DirectSubFrame.DSubframe;
        }
        case else {FatalError(__FILE__, __LINE__, "Invalid Coverage type");}
    }
    v_RxPositionInBitmap := (v_RxAbsoluteSubframe - v_OffsetIndicator) mod v_BitmapLength;
    v_Result := (p_SL_CommResourcePoolV2X_Rx.sl_Subframe_r14.bs20_r14[v_RxPositionInBitmap] == '1'B);
    return v_Result;
  }


4. Branches executed in test case 24.1.2

n/a

5. Execution Log Files

5.1. Qualcomm MDM 9150
The Qualcomm device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_24_1_2_Qualcomm.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

5. References
	[1]
	R5s180459
This archive comprises Execution log files, PICS/PIXIT files


